AD-A024 028

NUMERICAL METHOD FOR DETERMINATION OF MICROSTRIP INTERDIGITATED COUPLER DESIGN PARAMETERS

NAVAL WEAPONS CENTER

MARCH 1976

NWC TP 5843

Numerical Method for Determination of Microstrip Interdigitated Coupler Design Parameters

by
Donald D. Paolino
Electronic Warfare Department

MARCH 1976

Approved for public release; distribution unlimited.

Naval Weapons Center
CHINA LAKE, CALIFORNIA 93555

Naval Weapons Center

AN ACTIVITY OF THE NAVAL MATERIAL COMMAND

R. G. Freeman, III, RAdm., USN . . . . Commander
G. L. Hollingsworth . . . . Technical Director

FOREWORD

The purpose of this report is to present an approximate method for analysis of microstrip interdigitated directional couplers so that pertinent design parameters such as coupling coefficients, odd and even mode velocities, and characteristic impedance may be available to the design engineer.

The general method of analysis is reviewed, convergence of the method is investigated, and a computer program implementing this analysis is described in detail. Test data are presented in support of the program.

The program is used to generate design curves of ZOE and ZOO with W/H and S/H as parameters for substrate materials, alumina and quartz.

The work was conducted under the electromagnetic radiating source elimination (ERASE) program, AIRTASK A360-360E/008C/6W-AA23-001.

This report was reviewed for technical accuracy by Dixie D. Green and Joseph A. Mosko and is released at the working level.

Released by
D. J. RUSSELL, *Head*
Electronic Warfare Department
16 March 1976

Under authority of
G. L. HOLLINGSWORTH
Technical Director

NWC Technical Publication 5843

Published by . . . . Technical Information Department
Collation . . . . Cover, 17 leaves
First printing . . . . 145 unnumbered copies

REPORT DOCUMENTATION PAGE

READ INSTRUCTIONS BEFORE COMPLETING FORM

1. REPORT NUMBER NWC TP 5843	2. GOVT ACCESSION NO.	3. RECIPIENT'S CATALOG NUMBER
4. TITLE (and Subtitle) Numerical Method for Determination of Microstrip Interdigitated Coupler Design Parameters		5. TYPE OF REPORT & PERIOD COVERED A summary report
		6. PERFORMING ORG. REPORT NUMBER
7. AUTHOR(s) Donald D. Paolino		8. CONTRACT OR GRANT NUMBER(s)
9. PERFORMING ORGANIZATION NAME AND ADDRESS Naval Weapons Center China Lake, CA 93555		10. PROGRAM ELEMENT, PROJECT, TASK AREA & WORK UNIT NUMBERS A360-360E/008C/6W-AA23-001
11. CONTROLLING OFFICE NAME AND ADDRESS Naval Weapons Center China Lake, CA 93555		12. REPORT DATE March 1976
		13. NUMBER OF PAGES ~~32~~ 35
14. MONITORING AGENCY NAME & ADDRESS(if different from Controlling Office)		15. SECURITY CLASS. (of this report) UNCLASSIFIED
		15a. DECLASSIFICATION/DOWNGRADING SCHEDULE

16. DISTRIBUTION STATEMENT (of this Report)

Approved for public release; distribution unlimited.

17. DISTRIBUTION STATEMENT (of the abstract entered in Block 20, if different from Report)

18. SUPPLEMENTARY NOTES

19. KEY WORDS (Continue on reverse side if necessary and identify by block number)

Microstrip couplers
Interdigitated (Lange) couplers
Microstrip numerical design model
Coupler parameters

20. ABSTRACT (Continue on reverse side if necessary and identify by block number)

See back of form.

DD FORM 1 JAN 73 1473 EDITION OF 1 NOV 65 IS OBSOLETE
S/N 0102-014-6601

Numerical Method for Determination of Microstrip Interdigitated Coupler Design Parameters, by Donald D. Paolino. China Lake, Calif., Naval Weapons Center, February 1976. 32 pp. (NWC TP 5843, publication UNCLASSIFIED.)

A method of determining design parameters for the microstrip interdigitated (Lange) coupler is presented. Parameters investigated include coupling coefficients, odd and even mode velocities, and characteristic impedance. A general method of analysis is reviewed, convergence of the method is studied, and a computer program implementing this analysis along with supporting test data are presented. Test results indicate a good agreement between measured and predicted coupling and directivity for the interdigitated couplers.

CONTENTS

NOMENCLATURE

CED	Capacitance per unit length of N/2 strips to ground when all strips are held at the same potential and $\varepsilon = \varepsilon_0\varepsilon_r$
CEO	Capacitance per unit length of N/2 strips to ground when all strips are held at the same potential and $\varepsilon = \varepsilon_0$
COD	Capacitance per unit length of N/2 strips to ground when succeeding adjacent strips have equal magnitude opposite polarity potentials and $\varepsilon = \varepsilon_0\varepsilon_r$
COO	Capacitance per unit length of N/2 strips to ground when succeeding adjacent strips have equal magnitude opposite polarity potentials and $\varepsilon = \varepsilon_0$
ε	$\varepsilon_0\varepsilon_r$, permittivity of the substrate
ε_0	Permittivity of free space = 8.854×10^{-12} F/m
ε_r	Relative dielectric constant of the substrate material
λ_0	Free space wavelength at the design center frequency
H	Substrate thickness, inches
K	Voltage coupling coefficient = (ZOE - ZOO)/(ZOE + ZOO)
mks	Meter kilogram second
N	Number of interdigitated coupled strips, must be an even number
NWC	Naval Weapons Center
S/H	Strip gap (inch) to substrate thickness (inch) ratio
$\sigma(x)$	Charge density, C/m at point x
TEM	Transverse electromagnetic
V	Velocity of light in vacuum = 2.9979×10^8 m/sec
VE	Even mode velocity, m/sec
VO	Odd mode velocity, m/sec
W/H	Strip width (inch) to substrate thickness (inch) ratio
x	Field point
x'	Source point
ZOE	Even mode impedance, Ω
ZOO	Odd mode impedance, Ω

INTRODUCTION

In recent years, the microstrip interdigitated (Lange) coupler has received considerable attention from microstrip engineers.[1,2,3]

The Lange coupler is a quadrature hybrid used to obtain tight (≈ -3 dB) coupling in planar microstrip through normal etching procedures. This coupler may take the form shown in Figures 1a, 1b, or 1c.

A -3 dB coupler built from two coplanar coupled microstrips would be difficult to etch (on substrate thicknesses of interest) because the coupling gap would be too narrow. The Lange device has been built with great success up to and including J-band frequencies.

The present form of the widely used coupled strip program of Bryant and Weiss cannot be used to analyze interdigitated structures. The charge distribution for the two-strip case is entirely different from the interdigitated case. Ou's analysis[4] derives "equivalent" even and odd mode impedances for the coupler. The cross-sectional dimensions are then obtained by using Bryant and Weiss' data. The derivation of these equivalent impedances was based on equal mode velocities, when in fact they are unequal. Using this method, it would be impossible to observe the change in impedances when, for example, all strip widths and strip gaps are unequal; this could occur during the etching process. This paper will present an approximate method that overcomes these difficulities and allows design information to be easily obtained.

---

[1] Lange, J. "Interdigitated Stripline Quadrature Hybrid," IEEE Trans. *Microwave Theory and Techniques*, (Correspondence), Vol. MTT-17 (December 1969), pp. 1150-51.

[2] Waugh, R., and D. Lacombe. "Unfolding the Lange Coupler," IEEE Trans. *Microwave Theory and Techniques*, (Short papers), Vol. MTT-20 (November 1972), pp. 777-79.

[3] Miley, J. "Looking for a 3 to 8 dB Microstrip Coupler," *Microwaves*, Vol. 13 (March 1974), pp. 58-62.

[4] Ou, W. "Design Equations for an Interdigitated Directional Coupler," IEEE Trans. *Microwave Theory and Techniques*, (Short papers), Vol. MTT-23 (February 1975), pp. 253-55.



FIGURE 1. Microstrip Interdigitated Coupler Geometries.

The analysis here employs numerical methods to determine the even and odd mode charge distributions on the strips and the resulting capacitance from which the impedance and coupling parameters are derived. Green's function techniques are used.

MATHEMATICAL MODEL FOR INTERDIGITATED COUPLERS

The interdigitated directional coupler may be viewed as a multiconductor transmission line of N (N must be even) elements (not including the ground plane). Lossless TEM (static) analysis will be used to obtain all results. The conductors will be assumed infinitely thin. Coupling parameters may be determined by an analysis of the cross section of the lines. For the case of a Lange coupler, N = 4.

On any TEM N wire transmission line, N orthogonal modes can exist. For directional coupler use, only two modes are of interest: (1) the even mode where all conductors are held at the same potential, and (2) the odd mode where succeeding adjacent conductors have equal magnitude and opposite polarity potentials applied. The nondegenerate mode possibilities for the four-strip Lange coupler are shown in Figures 2a through 2d.

The wire bonds serve two important functions:

1. Mode suppression of mode types 2 and 4

2. Connection to input and output ports.

It is required to find the capacitance to ground of N/2 strips (since N/2 strips are connected by wire bonds) for the even and odd mode excitations, both with $\varepsilon = \varepsilon_0 \varepsilon_r$ (dielectric case) and with $\varepsilon = \varepsilon_0$ (air case). All coupling parameters, impedances, and mode velocities then may be determined from these quantities.

METHOD OF ANALYSIS

All strips are excited with 1-volt potentials (even mode case) with $\varepsilon = \varepsilon_0 \varepsilon_r$, and the charge distribution and total capacitance to



FIGURE 2. Nondegenerate Mode Configuration for Interdigitated Coupler (N = 4). (Note that mode types 2 and 4 are virtually eliminated because the potentials on those strips connected by wire bonds are forced to be the same.)

ground of N/2 strips are determined. This quantity is denoted CED. The same procedure is done with $\varepsilon = \varepsilon_0$ and CEO is determined. The odd mode air and dielectric capacitances are found and are denoted COD and COO. Note that since charge equals potential times capacitance and since the potential is 1 volt, the total charge is numerically the total capacitance.

The coupling parameters are:

1. ZOE = 1./V/(CEO CED)$^{1/2}$ even mode impedance, Ω (1)
2. ZOO = 1./V/(COD COO)$^{1/2}$ odd mode impedance, Ω (2)
3. K = (ZOE − ZOO)/(ZOE + ZOO) voltage coupling coefficient (3)
4. VE = V/(CED/CEO)$^{1/2}$ m/sec even mode velocity (4)
5. VO = V/(COD/COO)$^{1/2}$ m/sec odd mode velocity (5)

with V = speed of light in vacuum = 2.9979×10^8 m/sec.

To determine the charge distribution, each strip is subdivided into a number of substrips. Each substrip is taken to be a line charge and if the subdivision is small enough, the charge density is considered to be a constant over that substrip.

With reference to Figure 3, the potential is related to the charge density through an integral equation:

$$V(x) = \int_S G(x,x')\sigma(x')dx' \tag{6}$$

where S is a contour along which the equivalent source distributions lie and $\sigma(x')$ is the charge density at point x'.

The Green's function in Equation 6 is given by[5]:

$$G(x,x') = \left[2\pi\varepsilon_o(1+\varepsilon_r)\right]^{-1} \sum_{k=0}^{\infty}(-1)^k \left(\frac{\varepsilon_r - 1}{\varepsilon_r + 1}\right)^k \ln\left[\frac{(x-x')^2 + \left[2H(1+k)\right]^2}{(x-x')^2 + (2kH)^2}\right] \tag{7}$$

---

[5] Weeks, W. "Calculation of Coefficients of Capacitance of Multiconductor Transmission Lines in the Presence of a Dielectric Interface," IEEE Trans. *Microwave Theory and Techniques*, Vol. MTT-18 (January 1970), pp. 35-43.



FIGURE 3. Coordinate System and Strip Subdivision for Case N = 2 Substrips per Strip. (Primed coordinates--source points; unprimed coordinates--field points.)

x' will be denoted the "source point", and x will be denoted the "field point". The source point will be at the edge of a substrip and the field point will be at the center of a substrip. This is done to avoid numerical singularities in the Green's Function.

The moment method implementation[6] of a solution for σ(x') will now be described. For the purposes of simplicity in illustration, two substrips per strip will be assumed. Approximation of the integral in Equation 6 by a sum yields the discrete version of Equation 6.

$$\sum_{j=1}^{8} G(x_i, x_j)\sigma_j = V_i \qquad i = 1,8$$

in matrix form:

$$\begin{pmatrix} D_{11} & D_{12} & D_{13} & D_{14} & \cdot & \cdot & \cdot & D_{18} \\ \cdot & & & & & & & \cdot \\ \cdot & \cdot & & & & & & \cdot \\ \cdot & & \cdot & & & & & \cdot \\ \cdot & & & & & & & \cdot \\ \cdot & & & \cdot & & & & \cdot \\ \cdot & & & & & & & \cdot \\ D_{81} & \cdot & \cdot & \cdot & \cdot & \cdot & \cdot & D_{88} \end{pmatrix} \begin{pmatrix} \sigma_1 \\ \sigma_2 \\ \sigma_3 \\ \sigma_4 \\ \sigma_5 \\ \sigma_6 \\ \sigma_7 \\ \sigma_8 \end{pmatrix} = \begin{pmatrix} V_1 \\ V_2 \\ V_3 \\ V_4 \\ V_5 \\ V_6 \\ V_7 \\ V_8 \end{pmatrix} \qquad (8)$$

[6] Farrar, A., T. Adams. "Matrix Methods for Microstrip Three Dimensional Problems," IEEE Trans. *Microwave Theory and Techniques*, Vol. MTT-20, (Aug 1972) p. 497.

For the even mode case, all the voltages are +1, for the odd mode case, $V_1 = V_2 = V_5 = V_6 = +1$ volt and $V_3 = V_4 = V_7 = V_8 = -1$ volt.

D_{11} represents the contribution of the potential on substrip 1 due to a line charge on substrip 1, D_{12} represents the contribution of potential on substrip 1 due to a line charge on substrip 2 and so on.

Equation 8 must be solved four times; for the even mode air and dielectric cases, and for the odd mode air and dielectric cases.

The total capacitance for a case, i.e., CED, CEO, COD, or COO is the sum of the charge densities on alternate strips.

Let W be the width of a strip, and S be the spacing between strips. For the case when there are NSTRIP substrips per strip, each substrip will be W/NSTRIP wide. Denote this quantity by DELX.

The first source point is:

$$x_1' = -S/2 - S(N/2 - 1) - WN/2 \tag{9}$$

in the coordinate reference system of Figure 3. The first field point is:

$$x_1 = x_1' + DELX/2 \tag{10}$$

Both x_1 and x_1' are used in the computation of D_{11} in equation 8. After D_{11} is computed and stored, the next field point is calculated by adding DELX to equation 10 and D_{21} is computed. This process continues until D_{81} is formed. At that time, x_1' is incremented by DELX and the field point coordinate is reset to the initial value of equation 10. The process continues until the entire matrix is filled. Note that when the source or field point was computed on the last substrip of each strip, the next source or field point is obtained by adding S + DELX to the last respective source or field coordinate instead of adding DELX alone.

Observe that $D_{ij} \neq D_{ji}$. This is a consequence of picking source coordinates disjoint from the field point coordinates to avoid the singularity term in Green's function (equation 7) when $i = j$ (corresponding to $x = x'$) and $k = 0$. This technique has been used to approximately solve field problems[7], and it will be shown in the next section that this practice is valid by studying convergence of the method.

A computer program implementing the analysis was written in Fortran and implemented on the NWC Univac 1110. The listing of this program and its detailed description is included in the Appendix.

---

[7] Farrar, A., and T. Adams. "Correction to Computation of Lumped Microstrip Capacities by Matrix Method - Rectangular Sections and End Effect, IEEE Trans. *Microwave Theory and Techniques*, Vol. MTT-20 (April 1972), p. 294.

CONVERGENCE OF THE NUMERICAL METHOD

Figure 4 shows convergence data for tightly coupled (-3 dB) and loosely coupled (-5.78 dB) cases.



FIGURE 4. Convergence of Coupling Coefficients Versus Number of Substrips per Strip.

The coupling coefficient is plotted versus the number of substrips. The number of substrips range from 1 to 18. The convergence is of a damped oscillatory nature. Note that the coupling coefficient is significantly different for an even number of substrips than an odd number. This behavior is primarily due to choosing disjoint source and field points. If the singularity is rigorously treated, it can be shown that convergence is monotonic.

In the next section, these two cases will be studied in light of actual test data.

For the -3 dB case, there is little difference (0.0169 dB) in taking the 10-substrip solution over the 18-substrip case as will be shown.

For the case of 10 substrips, a coupling coefficient of -3.0589 dB is obtained corresponding to a W/H of 0.112 and an S/H of 0.08 (see Figure 11).

For the 18-substrip case, a coupling coefficient of -3.0420 dB is obtained and interpolation on the 10-substrip chart yields a W/H of 0.1100 and an S/H of 0.0805. For a 25-mil substrate, the difference in gap spacings and in line widths is of the order of a hundredth of a mil, a dimension so small that it cannot be controlled by normal etching procedures.

To ascertain that the program converges to the right coupling coefficient, the degenerate two-strip case was run and the results typically agreed with those of Bryant and Weiss to within 0.2% for the same number of substrips used (20).

Accuracy of the method should be unaffected by a choice of parameters S, H, or ε_r. However, as W/H is increased for a fixed number of subdivisions, accuracy will diminish. For this reason, structures with large W/H ratios should not be analyzed by this technique. This is not a practical limitation, since for the range of useful coupling coefficients and characteristic impedances, the interdigitated strip widths are narrow enough so that the 10-substrip subdivision is adequate.

TEST DATA

The purpose of this section is to compare the accuracy of the computer program results against actual interdigitated device test data obtained at this laboratory.

The first case to be considered is a coupler built to the dimensions found in the reference of Footnote 2. These are: W/H = 0.112 and S/H = 0.08.

The measured results for insertion loss are shown in Figure 5. The coupler is very close to a -3 dB coupler since there is virtually an equal power split at the design center frequency of 2 GHz.

Inputting the measured dimensions of this coupler into the computer program (using 10 substrips per strip) yields the results shown in Figure 6. The coupling coefficient is -3.0589 dB. Footnote 8 also reports data on a coupler etched to the same cross-sectional dimensions of Footnote 2 and is in perfect agreement with the data presented here.

---

[8] Naval Weapons Center. *A Microstrip Interdigitated 3 dB Coupler Operational Over Octave Plus Bandwidths*, by D. R. Bowling, et al. China Lake, Calif., NWC, 1973. (NWC TM 3524-146-73).



FIGURE 5. Measured Data, -3 dB Microstrip Interdigitated Coupler.

W/H = 0.112000

S/H = 0.800000E-01

ZOE, OHMS	ZOO, OHMS	COUPLING COEFF, dB	CHARACTERISTIC IMPEDANCE, OHMS
115.08592	20.054604	-3.0589046	48.037384

EFFKE	EFFKO	COUPLING COEFF	VEVEN M/SEC	VODD M/SEC
6.5679860	5.5530300	0.70316100	0.11097708E+09	0.12720862E+09

FIGURE 6. Computed Coupling Parameters 3 dB Microstrip Interdigitated Coupler.

Figure 7 shows test data for a device which is undercoupled. Averages of the line widths and gap widths were taken across the center of the coupler in the hope of inputting these to the computer program and predicting the 0.4 dB coupling unbalance at the design center frequency of 2 GHz.

This device had an average W/H of 0.103336 and an average S/H of 0.087600.

Figure 8 shows the results of the analysis of the program LANG, indicating a coupling coefficient of -3.198 dB. The even and odd mode impedances and dielectric constants were the input data to a microstrip analysis program[9] developed at NWC which models microstrip directional coupler insertion loss and isolation as a function of frequency. The differences between the insertion loss for the coupled port and the direct port at the design center frequency is shown in Figure 8 as 0.4 dB, exactly what was measured.

Figure 9 shows a more loosely coupled case (-5.78 dB). The isolation of the device is shown on the same scale as the insertion loss. The measured average cross-sectional parameters were: W/H = 0.14200 and S/H = 0.30264.

The computer printout (Figure 10) from program LANG shows the coupling parameters. The even and odd mode dielectric constants and impedances were input to the analysis program of Footnote 9 and the frequency response through the coupled and direct ports is printed along with the isolation data.

The predicted response is plotted on Figure 9 as a series of *s. In particular, the coupling unbalance at midband was measured as 5 dB; the computer printout predicts an unbalance of 4.921 dB. There also is good agreement between the predicted isolation and the measured one. The slight offset between measured and computed response can be attributed to connector and line losses, along with wire bond effects, none of which were computer modeled.

---

[9] Naval Weapons Center. *Solutions to Some Problems in Microstrip Filter Design,* by Donald D. Paolino. China Lake, Calif., NWC (NWC Technical Memorandum 2669, February 1976).



FIGURE 7. Undercoupled Interdigitated -3 dB Microstrip Coupler.

W/H= 103336   S/H= 876000E-01

ZOE (OHMS)	ZOO (OHMS)	COUPLING COEFF DB	CHARACTERISTIC IMPEDANCE(OHMS)
116 50389	21 210693	-3 1983614	49 710449

EFFKE	EFFKO	COUPLING COEFF	VEVEN M/SEC	VODD M/SEC
6 5566902	5 5540857	69196153	11707787E+09	12720696E+09

****

DIRECT PORT

FGHZ	RTLSS	VSWR	IN LSS	OUTPHS
2 00	-29 522	1 07	-2 829	-91 37

COUPLED PORT

FGHZ	RTLSS	VSWR	IN LSS	OUTPHS
2 00	-29 522	1 07	-3 221	-1 25

FIGURE 8. Coupling Parameters and Computer Predicted Direct and Coupled Port Midband Frequency Response, Undercoupled Hybrid.



FIGURE 9. Predicted and Measured Data, Loosely Coupled Case.

DESIGN CURVES

A series of design curves (Figures 11 and 12) now will be presented for the two substrate materials, alumina and quartz (bulk dielectric constants of $\varepsilon_r = 10.1$ and $\varepsilon_r = 3.78$, respectively). ZOE is plotted versus ZOO with W/H and S/H as parameters.

Alumina is a common substrate material, but its high dielectric constant makes designs above J-band difficult due to the small size of the components. Quartz is a good material for higher frequency applications since it can be polished to an optical finish and thus has lower loss characteristics than alumina. Also the fine finish will allow narrow gaps to be etched more easily. Quartz has a relatively constant bulk dielectric constant versus frequency and its low dielectric constant permits components to be built which are less subject to dimensional tolerances.

To use the curves, the voltage coupling coefficient K and characteristic impedance ZO (usually 50 Ω) must be known.

The odd mode impedance is determined by:

$$ZOO = ZO\ ((1 - K)/(1 + K))^{1/2} \qquad (11)$$

The even mode impedance is then:

$$ZOE = ZO^2/ZOO \qquad (12)$$

W/H= 142000

S/H= .302640

ZOE (OHMS)	ZOO (OHMS)	COUPLING COEFF DB	CHARACTERISTIC IMPEDANCE(OHMS)
4 410446	38 488892	-6. 1022434	49. 038361

EEFFE	EEFFO	COUPLING COEFF	VEVEN M/SEC	VODD M/SEC
6 [illegible]10675	5 5720835	49532223	11420205E+09	.12700136E+09

****

DIRECT PORT

F GHZ	RTLSS	VSWR	IN/LSS	OUTPHS	ISO
1 50	-31 599	1 05	-1 096	-71. 94	-26. 30
1 60	-30 770	1 06	-1 150	-76. 07	-25. 78
1 70	-29 993	1 07	-1 192	-80 16	-25. 25
1 80	-29 261	1 07	-1 221	-84. 22	-24 73
1 90	-28 566	1 08	-1 236	-88 26	-24. 19
2 00	-27 905	1 08	-1 238	-92 30	-23. 65
2 10	-27 274	1 09	-1 227	-96 35	-23 10
2 20	-26 668	1 10	-1 203	-100 42	-22. 53
2 30	-26 088	1 10	-1 166	-104 52	-21. 96
2 40	-25 530	1 11	-1. 117	-108 67	-21. 38
2 50	-24 995	1 12	-1. 058	-112. 88	-20. 80

COUPLED PORT

F GHZ	RTLSS	VSWR	IN/LSS	OUTPHS	ISO
1 50	-31 599	1 05	-6 575	18. 10	-26. 30
1 60	-30 770	1 06	-6 398	13. 99	-25 78
1 70	-29 993	1 07	-6 270	9. 93	-25. 25
1 80	-29 261	1 07	-6 189	5. 89	-24 73
1 90	-28 566	1 08	-6 153	1. 88	-24 19
2 00	-27 905	1 08	-6 159	-2. 12	-23. 65
2 10	-27 273	1 09	-6 210	-6. 12	-23. 10
2 20	-26 668	1 10	-6 305	-10. 13	-22. 53
2 30	-26 088	1 10	-6 447	-14 17	-21. 96
2 40	-25 530	1 11	-6 640	-18 24	-21. 38
2 50	-24 995	1 12	-6 887	-22. 35	-20. 80

FIGURE 10. Computer Predicted Coupling Data and Frequency Response, Loosely Coupled Case.



FIGURE 11. Interdigitated Coupler Design Curves for Alumina (ε_r = 10.1).



FIGURE 12. Interdigitated Coupler Design Curves for Quartz ($\varepsilon_r = 3.78$).

The point defined by a ZOE, ZOO pair will intersect a curve of constant W/H and another curve of constant S/H. These define the cross-sectional dimensions of the coupler. The quarter wavelength is determined by:

$$0.5\lambda_0/((\mathrm{EFFKE})^{1/2} + (\mathrm{EFFKO})^{1/2}) \qquad (13)$$

where λ_0 is the free space wavelength at the design center frequency and EFFKE and EFFKO are the even and odd mode dielectric constants. These are determined by inputting W and S into program LANG to obtain EFFKE and EFFKO.

A line of constant ZO = 50 Ω is plotted on all design curves. Thus, it only is necessary to find either ZOE or ZOO and locate the intersection of the impedance with the constant ZO line to obtain the physical dimensions. The range of coupling coefficients shown is from -6 to about -2 dB for a 50 Ω characteristic coupler impedance.

CONCLUSIONS

For the model employed in this report, quite good agreement between measured and predicted coupling and directivity has been obtained for interdigitated couplers.

An accurate model would include such effects as dielectric and conductor losses, finite strip thickness, wire bond effects, radiation and dispersion, and the proper treatment of the Green's function singularity.

For higher frequency coupler operation, (I-band and above) wire bond effects must be modeled to accurately predict isolation and coupling data. Perhaps these may be modeled as lumped discontinuities in the transmission lines.

It is possible to include the effects of finite conductor thickness, and program LANG may be modified to include the effect of unequal gap spacings and line widths.

It is hoped that this report will serve a useful purpose by providing design data for building devices, as well as laying a foundation from which more accurate models and theories of the interdigitated directional coupler may be brought forth.

Appendix

LISTINGS AND DETAILED DESCRIPTION OF COMPUTER PROGRAM

MAIN PROGRAM LANG

The main program has a capacity to handle 20 substrips per strip for the N = 4 case.

The program is conversational and asks for the following input data in lines 8 through 25:

1. Substrate bulk dielectric constant, substrate height (inch), and the number of coupled strips.

2. If N = 1, only the strip width is read in line 17. For N = 2 or more, the gap spacing is entered in line 14.

3. Number of substrips into which each strip will be divided. It is strongly recommended that this be an even number.

4. Logical unit number for output. This will usually be 6.

Note that free-field format is used. The free-field format is indicated by () in line 10.

In lines 29 through 39, the input data are echoed on the computer printout as an aid to determine that the data have, in fact, been entered correctly and to document the particular computer run.

In lines 43 through 48, the line widths, gap widths, and substrate height are converted from English units to mks (meter kilogram second) units and the W/H and S/H ratios are computed.

The stepping routine, indicated in line 50, begins the heart of the program. Certain integer variables are defined which will serve as DO loop terminators. These are:

1. IEND = (Number of substrips per strip) (number of coupled strips)

2. I2S = IEND^2

IEND is the total number of elements in a column or row in the matrix of Green's function entries. I2S is the total number of elements in Green's function matrix.

The basic increment for the source and field point coordinates is defined in line 54 as DELXJ. The first source point is defined in line 55 corresponding to equation 9 and the first field point is defined in line 62 after equation 10. In case N = 1, the first source point is obviously -GW/2 and this is taken care of in line 56. In lines 58 and 59, two counters are set. The first, KOUNT3, controls the numbering of Green's function entries and the second, KOUNT2, controls when the source point will be incremented by DELXJ + GS instead of incrementing by DELXJ. KOUNT2 is reset every NSTRIP times through the 100 DO loop. Note that loop 100 controls the source coordinate incrementation and loop 110 controls the field point incrementation. As I is incremented with J fixed, the Jth column of Green's function matrix in equation 8 is filled. The terminal indices on these loops are IEND, since there are IEND rows and IEND columns in Green's function matrix D_{ij}. KOUNT1 is defined in line 61 and performs the same function on the field coordinate as KOUNT2 does for the source coordinate. The call to subroutine GREEN in line 67 occurs for the dielectric case ($\varepsilon = \varepsilon_0 \varepsilon_r$). The appropriate matrix entry is stored in the array DIJD. Note that D_{ij} is stored columnwise in DIJD. This is required for use in subroutine GELG to be described. The air case ($\varepsilon = \varepsilon_0$) occurs in line 69 and Green's function entry is placed in array DIJAIR. Green's function entry for both the air and dielectric cases is returned from subroutine GREEN as the variable VALG. (See the description of subroutine GREEN for details about Green's function calculation.) On line 65 is a test to determine if the program is on a diagonal of the D_{ij} matrix other than the first one. This is done to save time in program execution since Green's function evaluation is the most time-consuming program segment. All diagonal terms have the same numerical value. Statement 70 detects when you are on the appropriate diagonal and sets VALG to the first value D_{11} (either DIJD(1) or DIJAIR(1)). Then there is no need to call subroutine GREEN in this case and the IF statements in lines 66 and 71 transfer directly to the statements which fill the D_{ij} matrix. In line 74, KOUNT3 is incremented, and if incrementation of the field point is not going to include GS (determined by the test in line 75), the field point is further incremented across the strip by DELXJ to the next subdivision, KOUNT1 is incremented, and loop 110 continues. If incrementation includes GS, line 75 will cause a transfer to line 79 where GS is appended, counter KOUNT1 is reset to 1, and loop 110 continues. Lines 82 through 88 perform a similar function on the source point incrementation. When all I2S entries have been determined for DIJAIR and DIJD, the system of equation 8 is ready to be solved.

The next series of statements set up the appropriate right-hand-side vector of Equation 8, representing the even and odd mode voltage excitations.

Loop 120 will fill the first IEND locations of array R and array RD (a duplicate array to R) with the even mode excitation (all +1s). This filling is done through an array RHS returned from subroutine VEC. (See the description of subroutine VEC for programming details.) The integer variable IMODE (line 95) controls the even or odd mode excitation cases. IMODE = 0 for even mode and IMODE = 1 for odd mode excitation. A duplicate array RD is made because when the solution routine GELG is called, the input right-hand-side vector R must be input but on return, R is overwritten with the solution vector containing the charge distribution. This routine GELG was called for the dielectric case, but the vector R is needed again for the air case so to avoid regenerating R, it is stored as RD. For the case N = 1, line 102 will avoid generating the odd mode voltage distribution in arrays R and RD and transfer immediately to the equation solving routine GELG. For the multistrip case, lines 106 through 112 will fill R and RD with the odd mode excitation voltages.

At this point, the matrix of equation 8 is ready to be solved by GELG. (See the description of GELG for further details.) Line 119 will solve for the even and odd mode charge density distributions for the dielectric Green's function. The integer variable IN in the call to GELG in line 119 is the number of right-hand-side vectors to be solved. For the multistrip case, IN = 2 (one for the even mode and one for the odd mode) and this value was set in line 6. For the case of N = 1, only one mode (even mode) need be solved and in that case, IN was defined to be 1 in line 33. Note that the first IEND locations of the solution vector contain the charge distribution on the strips for the even mode dielectric case and the next IEND locations contain the odd mode charge distribution for the dielectric case.

Lines 129 to 141 perform the summing of the charge on alternate strips to obtain CED, the even mode dielectric capacitance. N/2 strips must be summed and this quantity is defined as NS2 in line 130. In case N = 1 or N = 2, only one strip need be summed and this case is taken care of in line 131. The running sum of capacitance is initialized to 0 by line 129. IS and IE are defined in lines 132 and 133. These are the initial and terminal parameters for the loop 140. Loop 140 sums the array R which contains the even mode charge distribution. The first NSTRIP locations are summed and when this is done, lines 138 and 139 define new initial and terminal parameters IS and IE. Thus, the next partial sum will start with an index skipping the next two NSTRIP cells from the previously defined value of IS.

CED is defined and if N = 1, there is no odd mode processing to take place so control is passed to statement 210 where the air case is solved. For the odd mode case, the solution to the charge distribution is found in location (IEND + 1) through (2 IEND) of array R. The operation of lines 146 through 156 is based on the same principle as lines 132 to 141. After loop 160 is completed, COD (the odd mode dielectric capacitance is defined in line 156.

The even mode and odd mode air cases are solved by inputting the DIJAIR array to the equation solving routine GELG. Vector RD is used as the array containing the even and odd mode excitations, and similar processing for that used to obtain CED and COD is used to obtain CEO and COO.

After these quantities are determined, the impedance and coupling parameters are found by means of the relations of Equations 1 through 5.

```
      DIMENSION RD(160),DIJD(7000),DIJAIR(7000),R(160)
      REAL RHS(80)
C
C ENTER INPUT DATA
C
      IN=2
      GS=0.0
      WRITE(6,10)
10    FORMAT(1H ,'ENTER SUBSTRATE DIEL. CONST., HEIGHT, AND # STRIPS'/)
101   FORMAT()
      READ101,ER,H,N
      IF(N.GT.1)WRITE(6,20)
20    FORMAT(1H ,'ENTER GAP SPACING, IN. BETWEEN STRIPS'/)
      IF(N.GT.1)READ101,GS
      WRITE(6,30)
30    FORMAT(1H ,'ENTER WIDTH OF STRIPS, IN.'/)
      READ101,GW
      E0=8.854E-12
      PI=3.141592654
      WRITE(6,40)
40    FORMAT(1H ,'GIVE # OF SUBSTRIPS DIVIDING EACH STRIP')
      READ 101,NSTRIP
      WRITE(6,41)
41    FORMAT(1H ,'WRITE LOGICAL UNIT FOR OUTPUT'/)
      READ 101,LU
C
C WRITE OUT DATA
C
      WRITE(LU,50)ER,H,N
50    FORMAT(1H ,'SUBSTRATE DIELECTRIC CONSTANT=' G20.8,
     ./' SUBSTRATE HEIGHT =',G20.8,' IN.'/
     .,I5,' STRIP COUPLER'/)
      IF(N.EQ.1)IN=1
      IF(N.GT.1) WRITE(LU,51)GW,GS
      IF(N.EQ.1) WRITE(LU,5011)GW
51    FORMAT(1H ,'STRIP WIDTH=',G20.8,' IN.',' STRIP GAP=',G20.8,' IN.')
5011  FORMAT(1H ,'STRIP WIDTH=',G20.8)
      WRITE(LU,5012)NSTRIP
5012  FORMAT(1H ,'# OF SUBSTRIPS PER STRIP=',I5/)
C
C CONVERT ALL DIMENSIONS TO METRIC(MKS SYSTEM)
C
      CP=39.3700787
      GW=GW/CP
      GS=GS/CP
      H=H/CP
      WTOH=GW/H
      STOH=GS/H
C
C STEPPING ROUTINE
```

```
C
      IEND=NSTRIP*N
      I2S=IEND*IEND
      DELXJ=GW/(FLOAT(NSTRIP))
      XJS=-GS/2.-GS*((FLOAT(N/2-1)))-GW*(FLOAT(N/2))
      IF(N.EQ.1)XJS=-GW/2.
      XJ=XJS
      KOUNT3=1
      KOUNT2=1
      DO 100 J=1,IEND
      KOUNT1=1
      XI=XJS+DELXJ*.5
      DO 110 I=1,IEND
      E=ER*EO
      IF(I.EQ.J.AND.I.NE.1)VALG=DIJD(1)
      IF(I.EQ.J.AND.I.NE.1)GO TO 500
      CALL GREEN(EO,E,PI,XJ,XI,H,VALG,LU)
  500 DIJD(KOUNT3)=VALG
      E=1.000001*EO
      IF(I.EQ.J.AND.I.NE.1)VALG=DIJAIR(1)
      IF(I.EQ.J.AND.I.NE.1)GO TO 501
      CALL GREEN(EO,E,PI,XJ,XI,H,VALG,LU)
  501 DIJAIR(KOUNT3)=VALG
  991 KOUNT3=KOUNT3+1
      IF(KOUNT1.EQ.NSTRIP)GO TO 90
      KOUNT1=KOUNT1+1
      XI=XI+DELXJ
      GO TO 110
   90 XI=XI+DELXJ+GS
      KOUNT1=1
  110 CONTINUE
      IF(KOUNT2.EQ.NSTRIP) GO TO 91
      KOUNT2=KOUNT2+1
      XJ=XJ+DELXJ
      GO TO 100
   91 XJ=XJ+DELXJ+GS
      KOUNT2=1
  100 CONTINUE
C
C PREPARE TO INVERT MATRICES OF DIJ'S
C
C EVEN MODE CASE
C
      KOUNT4=1
      IMODE=0
      CALL VEC(IMODE,N,NSTRIP,RHS)
      DO 120 I=1,IEND
      R(KOUNT4)=(RHS(I))
      RD(KOUNT4)=R(KOUNT4)
      KOUNT4=KOUNT4+1
  120 CONTINUE
      IF(N.EQ.1)GO TO 131
C
C ODD MODE CASE
C
      IMODE=1
      CALL VEC(IMODE,N,NSTRIP,RHS)
      DO 130 I=1,IEND
      R(KOUNT4)=(RHS(I))
      RD(KOUNT4)=R(KOUNT4)
      KOUNT4=KOUNT4+1
  130 CONTINUE
C
```

```
C NOW PROCESS THE DIELECTRIC GREENS FUNCTION
C MATRIX R WILL BE NEEDED AGAIN TO PROCESS THE AIR GREENS FUNCTION.
C SINCE R IS DESTROYED IN GELG, A DUPLICATE R CALLED RD WAS DEFINED
C ABOVE
C
131   CALL GELG(R,DIJD,IEND,IN,1.E-5,IER)
C
C R NOW CONTAINS THE SOLUTION TO THE EVEN MODE CHARGE
C DISTRIBUTION IN THE FIRST IEND LOCATIONS. FROM IEND+1 TO 2*IEND
C ARE FOUND THE SOLUTION TO THE ODD MODE DIELECTRIC CASE.
C
C NOW SUM THE CHARGES ON N/2 ALTERNATE FINGERS TO OBTAIN THE
C CAPACITANCE PER UNIT LENGTH OF THE STRIPS IN EVEN AND ODD MODES
C
C     EVEN MODE DIELECTRIC CASE
      SUM=0.0
      NS2=N/2
      IF(N.EQ.1.OR.N.EQ.2)NS2=1
      IS=1
      IE=NSTRIP
      DO 150 II=1,NS2
      DO 140 I=IS,IE
      SUM=SUM+R(I)
140   CONTINUE
      IS=IS+NSTRIP*2
      IE=IE+2*NSTRIP
150   CONTINUE
      CED=SUM
      IF(N.EQ.1) GO TO 210
C
C NOW PROCESS THE ODD MODE DIELECTRIC CASE
C
      IS=IEND+1
      IE=IEND+NSTRIP
      SUM=0.0
      DO 160 II=1,NS2
      DO 165 I=IS,IE
      SUM=SUM+R(I)
165   CONTINUE
      IS=IS+NSTRIP*2
      IE=IE+2*NSTRIP
160   CONTINUE
      COD=SUM
C
C NOW PROCESS THE AIR CASE
C
210   CALL GELG(RD,DIJAIR,IEND,IN,1.E-5,IER)
C
C NOW SUM THE CHARGES TO OBTAIN CAPACITANCE AS IN THE LOOP 150
C
C EVEN MODE AIR CASE
      IS=1
      IE=NSTRIP
      SUM=0.0
      DO 220 II=1,NS2
      DO 230 I=IS,IE
      SUM=SUM+RD(I)
230   CONTINUE
      IS=IS+NSTRIP*2
      IE=IE+2*NSTRIP
220   CONTINUE
      CEO=SUM
C
```

```
C ODD MODE AIR CASE
C
        IF(N.EQ.1)GO TO 260
        IS=IEND+1
        IE=IEND+NSTRIP
        SUM=0.0
        DO 240 II=1,NS2
        DO 250 I=IS,IE
        SUM=SUM+RD(I)
250     CONTINUE
        IS=IS+NSTRIP*2
        IE=IE+2*NSTRIP
240     CONTINUE
        CCO=SUM
C
C NOW THE IMPEDANCES CAN BE FOUND
C
260     V=2.9979E+08
        ZOE=1./(V*SQRT(CED*CEO))
        IF(N.EQ.1)WRITE(LU,997)ZOE
997     FORMAT(1H ,'ZO=',G20.8)
        IF(N.EQ.1)GO TO 1000
        ZOO=1./(V*SQRT(COD*COO))
        COUP=(ZOE-ZOO)/(ZOE+ZOO)
        CUPDB=8.68588964*ALOG(COUP)
        EFFKE=CED/CEO
        EFFKO=COD/COO
        VE=V/SQRT(EFFKE)
        VO=V/SQRT(EFFKO)
        CHAR=SQRT(ZOE*ZOO)
        WRITE(LU,990)WTOH,STOH
990     FORMAT(1H ,11X,'W/H=',G15.8,18X,'S/H=',G15.8//)
        WRITE(LU,996)
996     FORMAT(1H ,6X,'ZOE',13X,'ZOO',10X,'COUPLING',7X,'CHARACTERISTIC')
        WRITE(LU,995)
995     FORMAT(1H ,5X,'(OHMS)',10X,'(OHMS)',8X,'COEFF. DB',6X,
       .'IMPEDANCE(OHMS)'/)
        WRITE(LU,998)ZOE,ZOO,CUPDB,CHAR
        WRITE(LU,993)
993     FORMAT(1H ,3X,'EFFKE',9X,'EFFKO',9X,'COUPLING',8X,'VEVEN',7X,'
       .VODD')
        WRITE(LU,992)
992     FORMAT(1H ,32X,'COEFF.',9X,'M/SEC',9X,'M/SEC'/)
        WRITE(LU,994)EFFKE,EFFKO,COUP,VE,VO
994     FORMAT(1H ,5G14.8//)
998     FORMAT(1H ,2G16.8,G16.8,G16.8//)
1000    END
```

SUBROUTINE GREEN

This routine computes the infinite series (equation 7) representing Green's function for the problem.

VALG represents the value of Green's function for a given XI, XJ, substrate thickness H, and relative permittivity, ε_r.

Loop 10 on line 5 has a maximum index of 200, so up to 200 terms in the sum can be computed. If the series did not converge after 199 terms, line 23 will cause a message to be printed to that effect. It was found that in all cases run so far, Green's function converged after about 60 terms for the case where $\varepsilon = \varepsilon_0\varepsilon_r$ and in about three terms for $\varepsilon = \varepsilon_0$.

A cumulative sum is obtained in line 18. When the loop index N is 1, a value PREV is set to the first value of the sum. Thereafter, the next value of SUM is compared to PREV and the convergence condition is that PREV and SUM differ in no less than five significant figures. If this condition is not met, the new value of PREV is the old value of SUM and a new SUM is computed. When the convergence condition is met, control is passed to statement 11 where VALG is defined and control is passed to the main program.

```
 1:      SUBROUTINE GREEN(EO,E,PI,XJ,XI,H,VALG,LU)
 2:      P=E+EO
 3:      R=(E-EO)/P
 4:      SUM=0.0
 5:      DO 10 N=1,200
 6:      NN=N-1
 7:      T1=(FLOAT(2*NN))*H
 8:      T1S=T1*T1
 9:      T2=(FLOAT(2*NN+2))*H
10:      T2S=T2*T2
11:      A=((XI-XJ)*(XI-XJ))
12:      ANUM=A+T2S
13:      DNOM=A+T1S
14:      RT=ANUM/DNOM
15:      IF(RT.LE.0.0)RT=1.E-38
16:      TERM1=ALOG(RT)
17:9     S1=TERM1
18:      SUM=SUM+S1*((-1.)**(NN))*(R**(NN))
19:      IF(N.EQ.1)PREV=SUM
20:      IF(N.EQ.1) GO TO 10
21:      IF(ABS(PREV-SUM).LT.1.E-5)GO TO 11
22:      PREV=SUM
23:      IF(N.EQ.199)WRITE(LU,20)
24:20    FORMAT(1H ,'GREENS FUNCTION DID NOT CONVERGE'/)
25:10    CONTINUE
26:11    VALG=(SUM)/(2.*PI*P)
27:      RETURN
28:      END
```

SUBROUTINE VEC

This subroutine sets up the right-hand-side vector in equation 8 prior to calling the simultaneous equation solving routine GELG.

The parameter IMODE in the calling sequence is a flag variable used to obtain the even or odd mode voltage excitations.

For the even mode, IMODE = 0 and for the odd mode, IMODE = 1.

When IMODE = 0, the GO TO statement in line 4 directs control to loop 110 whose job it is to fill the right-hand-side vector with all 1s and immediately return to the main program.

The case where IMODE = 1 is more complicated. Control is directed to line 10 where a counter called KOUNT is set to 1. This counter will be used to control if the current location in the RHS vector is going to be filled with a 1 or a -1. The first NSTRIP locations in RHS are +1s, the next NSTRIP locations are -1s and so on. When KOUNT reaches the value NSTRIP (number of substrips per strip), the counter is reset and another flag variable called IFLAG changes sign. This is done in lines 18 and 19. The first location in the next set of NSTRIP values is initialized and the loop continues. Note that when IFLAG changes sign so do the next NSTRIP locations of RHS. When all NSTRIP.N values of RHS have been loaded, return is to the main program.

```
 1:        SUBROUTINE VEC(IMODE,N,NSTRIP,RHS)
 2:        DIMENSION RHS(1)
 3:        IEND=N*NSTRIP
 4:        IF(IMODE.EQ.0)GO TO 100
 5:        IF(IMODE.EQ.1)GO TO 200
 6:100     DO 110 I=1,IEND
 7:        RHS(I)=1.0
 8:110     CONTINUE
 9:        RETURN
10:200     KOUNT=1
11:        IFLAG=1
12:        DO 210 I=1,IEND
13:        IF(KOUNT.GT.NSTRIP)GO TO 209
14:        IF(IFLAG.EQ.1)RHS(I)=+1.0
15:        IF(IFLAG.EQ.-1)RHS(I)=-1.0
16:        KOUNT=KOUNT+1
17:        GO TO 210
18:209     KOUNT=1
19:        IFLAG=-IFLAG
20:        IF(IFLAG.EQ.-1)RHS(I)=-1.0
21:        IF(IFLAG.EQ.+1)RHS(I)=+1.0
22:        KOUNT=KOUNT+1
23:210     CONTINUE
24:        RETURN
25:        END
```

SUBROUTINE GELG

This routine, taken from IBM System/360 Scientific Subroutine Package Version III,[10] is fully documented therein as well as in the comment cards of the listing.

```
      SUBROUTINE GELG(R,A,M,N,EPS,IER)
      DIMENSION R(1),A(1)
C SUBROUTINE GELG
C
C PURPOSE
C    TO SOLVE A GENERAL SYSTEM OF SIMULTANEOUS EQUATIONS
C USAGE
C  CALL GELG(R,A,M,N,EPS,IER)
C
C DESCRIPTION OF PARAMETERS
C
C   R  - THE M X N MATRIX OF RIGHT HAND SIDES. (DESTROYED)
C        ON RETURN R CONTAINS THE SOLUTION OF THE EQUATIONS
C   A  - THE M X M COEFFICIENT MATRIX. (DESTROYED)
C   M  - THE NUMBER OF EQUATIONS IN THE SYSTEM.
C   N  - THE NUMBER OF RIGHT HAND SIDE VECTORS.
C   EPS   - AN INPUT CONSTANT WHICH IS USED AS RELATIVE
C      TOLERANCE FOR TEST ON LOSS OF SIGNIFICANCE.
C   IER   - RESULTING ERROR PARAMETER CODED AS FOLLOWS
C   IER=0 NO ERROR.
C   IER=-1 NO RESULT BECAUSE OF M LESS THAN 1 OR PIVOT ELEMENT
C       STEP EQUAL TO 0
C   IER=K WARNING DUE TO POSSIBLE LOSS OF SIGNIFICANCE INDICATED AT ELIMINATION
C STEP K+1, WHERE PIVOT ELEMENT WAS LESS THAN OR EQUAL TO THE
C INTERNAL TOLERRNCE EPS TIMES ABSOLUTELY GREATEST ELEMENT OF MATRIX A
C
C REMARKS
C INPUT MATRICES R AND A ARE ASSUMED TO BE STORED COLUMNWISE
C IN M*N RESP. M*M SUCCESSIVE STORAGE LOCATIONS. ON RETURN
C SOLUTION MATRIX R IS STORED COLUMNWISE TOO.
C THE PROCEEDURE GIVES RESULTS IF THE NUMBER OF EQUATIONS M IS GREATER
C THAN 0 AND PIVOT ELEMENTS AT ALL ELIMINATION STOPS
C ARE DIFFERENT FROM 0. HOWEVER WARNING IER=N IF GIVEN INDICATES POS-
C SIBLE POSSIBLE LOSS OF SIGNIFICANCE. IN CASE OF A WELL SCALED MATRIX
C A AND APPROPRIATE TOLERANCE EPS, IER=K MAY BE INTERPRETED
C THAT  MATRIX A HAS THE RANK K. NO WARNING IS GIVEN IN CASE M=1.
C
      IF(M)23,23,1
C
C SEARCH FOR GREATEST ELEMENT IN MATRIX A
1     IER=0
      PIV=0.
      MM=M*M
      NM=N*M
      DO 3 L=1,MM
      TB=ABS(A(L))
```

10 International Business Machines. *IBM System/360 Scientific Subroutine Package Version III, Programmer's Manual.* IBM, 1968. Pp. 121-23, (Program Number 360A-CM-03X).

```
      IF(TB-PIV)3,3,2
2     PIV=TB
      I=L
3     CONTINUE
      TOL=EPS*PIV
C     A(I) IS PIVOT ELEMENT. PIV CONTAINS THE DABSOLUTE VALUE OF A(I)
C
C
C START ELIMINATION LOOP
      LST=1
      DO 17 K=1,M
C
C TEST ON SINGULARITY
      IF(PIV)23,23,4
4     IF(IER)7,5,7
5     IF(PIV-TOL)6,6,7
6     IER=K-1
7     PIVI=1./A(I)
      J=(I-1)/M
      I=I-J*M-K
      J=J+1-K
C I+K IS ROW INDEX, J+K COLUMN-INDEX OF PIVOT ELEMENT
C
C PIVOT ROW REDUCTION AND ROW INTERCHANGE IN RIGHT HAND SIDE R
      DO 8 L=K,NM,M
      LL=L+I
      TB=PIVI*R(LL)
      R(LL)=R(L)
8     R(L)=TB
C
C IS ELIMINATION TERMINATED
      IF(K-M)9,18,18
C
C COLUMN INTERCHANGE IN MATRIX A
9     LEND=LST+M-K
      IF(J)12,12,10
10    II=J*M
      DO 11 L=LST,LEND
      TB=A(L)
      LL=L+II
      A(L)=A(LL)
11    A(LL)=TB
C
C ROW INTERCHANGE AND PIVOT ROW REDUCTION IN MATRIX A
12    DO 13 L=LST,MM,M
      LL=L+I
      TB=PIVI*A(LL)
      A(LL)=A(L)
13    A(L)=TB
C
C SAVE COLUMN INTERCHANGE INFORMATION
      A(LST)=J
C
C ELEMENT REDUCTION AND NEXT PIVOT SEARCH
      PIV=0.
      LST=LST+1
      J=0
      DO  16 II=LST,LEND
      PIVI=-A(II)
      IST=II+M
      J=J+1
      DO 15 L=IST,MM,M
```

```
      LL=L-J
      A(L)=A(L)+PIVI*A(LL)
      TB=ABS(A(L))
      IF(TB-PIV)15,15,14
14    PIV=TB
      I=L
15    CONTINUE
      DO 16 L=K,NM,M
      LL=L+J
16    R(LL)=R(LL)+PIVI*R(L)
17    LST=LST+M
C END OF ELIMINATION LOOP
C
C
C BACK SUBSTITUTION AND BACK INTERCHANGE
18    IF(M-1)23,22,19
19    IST=MM+M
      LST=M+1
      DO 21 I=2,M
      II=LST-I
      IST=IST-LST
      L=IST-M
      L=A(L)+.5
      DO 21 J=II,NM,M
      TB=R(J)
      LL=J
      DO 20 K=IST,MM,M
      LL=LL+1
20    TB=TB-A(K)*R(LL)
      K=J+L
      R(J)=R(K)
21    R(K)=TB
22    RETURN
C
C
C ERROR RETURN
23    IER=-1
      RETURN
      END
```

REFERENCES

1. Lange, J. "Interdigitated Stripline Quadrature Hybrid," *Inst Elec Electron Eng, Trans, Microwave Theory and Techniques*, (Correspondence), Vol. MTT-17 (December 1969), pp. 1150-51.

2. Waugh, R., and D. Lacombe. "Unfolding the Lange Coupler," *Inst Elec Electron Eng, Trans, Microwave Theory and Techniques*, (Short papers), Vol. MTT-20 (November 1972), pp. 77-79.

3. Miley, J. "Looking for a 3 to 8 dB Microstrip Coupler," *Microwaves*, Vol. 13 (March 1974), pp. 58-62.

4. Ou, W. "Design Equations for an Interdigitated Directional Coupler," *Inst Elec Electron Eng, Trans, Microwave Theory and Techniques*, (Short papers), Vol. MTT-23 (February 1975), pp. 253-55.

5. Weeks, W. "Calculation of Coefficients of Capacitance of Multiconductor Transmission Lines in the Presence of a Dielectric Interface," *Inst Elec Electron Eng, Trans, Microwave Theory and Techniques*, Vol. MTT-18 (January 1970), pp. 35-43.

6. Farrar, A., T. Adams. "Matrix Methods for Microstrip Three Dimensional Problems," *Inst Elec Electron Eng, Trans, Microwave Theory and Techniques*, Vol. MTT-20 (Aug 1972), p. 497.

7. ------. "Correction to Computation of Lumped Microstrip Capacities by Matrix Method - Rectangular Sections and End Effect, *Inst Elec Electron Eng, Trans, Microwave Theory and Techniques*, Vol. MTT-20 (April 1972), p. 294.

8. Naval Weapons Center. *A Microstrip Interdigitated 3 dB Coupler Operational Over Octave Plus Bandwidths*, by D. R. Bowling, et al. China Lake, Calif., NWC, 1973. (NWC TM 3524-146-73).

9. ------. *Solutions to Some Problems in Microstrip Filter Design*, by Donald D. Paolino. China Lake, Calif., NWC, February 1976. (NWC Technical Memorandum 2669, publication UNCLASSIFIED.)

10 International Business Machines. *IBM System/360 Scientific Subroutine Package Version III, Programmer's Manual.* IBM, 1968. Pp. 121-23, (Program Number 360A-CM-03X).